The function `map`

```
suclist :: [Int] -> [Int]
suclist []     = []
suclist (x:xs) = suc x : suclist xs

sqrtlist :: [Float] -> [Float]
sqrtlist []     = []
sqrtlist (x:xs) = sqrt x : sqrtlist xs

map :: (a -> b) -> [a] -> [b]
map g [] = []
map g (x:xs) = g x : map g xs
```



```
suclist :: [Int] -> [Int]
suclist  = map suc
```

```
sqrtlist :: [Float] -> [Float]
sqrtlist = map sqrt
```

The function `filter`

```
dropEven :: [Int] -> [Int]
dropEven [] = []
dropEven (x:xs) | odd x     = x : dropEven xs
                | otherwise = dropEven xs

dropUpper :: [Char] -> [Char]
dropUpper [] = []
dropUpper (x:xs) | isLower x = x : dropUpper xs
                 | otherwise = dropUpper xs

filter :: (a -> Bool) -> [a] -> [a]
filter g [] = []
filter g (x:xs) | g x       = x : filter g xs
                | otherwise = filter g xs
```



```
dropEven :: [Int] -> [Int]
dropEven  = filter odd
```

```
dropUpper :: [Char] -> [Char]
dropUpper = filter isLower
```

The function `foldr`

```
sum :: Num a => [a] -> a
sum []     = 0
sum (x:xs) = x + sum xs
```

```
prod ::  Num a => [a] -> a
prod []     = 1
prod (x:xs) = x * prod xs
```

```
concat :: [[a]] -> [a]
concat []     = []
concat (x:xs) = x ++ concat xs
```

```
foldr :: (a -> b -> b) -> b -> [a] -> b
foldr g e []     = e
foldr g e (x:xs) = g x (foldr g e xs)
```



```
sum :: Num a => [a] -> a
sum  = foldr (+) 0
```

```
prod ::  Num a => [a] -> a
prod = foldr (*) 1
```

```
concat :: [[a]] -> [a]
concat = foldr (++) []
```